Measuring The Characteristic Impedance of Balanced Cables

- HP 4194A Application Information -



Introduction

The HP 4194A Impedance/Gain-Phase Analyzer can measure balanced cables (such as twisted pair cables) quickly and efficiently. This application information describes how to measure the characteristic impedance of balanced cables by using an open/short method which is useful for cable manufacturers and cable users (telecommunications, telephone, TV, computer, and instrument manufacturers).

Application Issue

Characteristic impedance is the most often used parameter for evaluating the transmission characteristics of cables. Measuring the characteristic impedance of a balanced cable is not as easy as it is for an unbalanced cable, though, because a balanced measurement circuit is required. In the past, the following techniques have been used.

* Impedance Bridge
Time consuming test and requires a highly skilled user.

* LCR Meter with a Balun
An LCR Meter's test terminals are unbalanced, so it is necessary to insert a balun between the LCR Meter measurement terminals and the test device (in this case a cable). This requires complicated calculations to compensate for the transmission errors in balun (a complicated compensation program to run on an external computer must and the measurement speed suffers accordingly.

Solutions Offered by the HP 4194A

* Easy to setup Balanced Measurements and High Measurement Speed
The HP 4194A has powerful compensation functions, so errors due to the balun can be easily compensated for, and measurement speed is not lost by having to make lengthy calculations to correct for the presence of a balun in the circuit.

* Wide Frequency/Measurement
The HP 4194A has a frequency range of 100Hz to 40MHz for impedance measurements (10kHz) to 100MHz with the HP 41941A/B), and a measurement range of 10mΩ to 100MΩ (0.1Ω to 1MΩ with the HP 41941A/B). This measurement range is wide enough to make open/short measurements of cables.

* Secondary Parameter Analysis
After a measurement is made, the measurement data can be used to calculate and display the characteristic impedance of the cable, and other secondary parameters.

* Gain-Phase and Additional Impedance Evaluation
The HP 4194A can also be used to determine other impedance parameters such as inductance, capacitance, and the dielectric constant of cable materials. The HP 4194A's Gain-Phase measurement function can be used to measure transmission characteristics such as cross-talk, attenuation, and delay time.

* Auto Sequence Program (ASP)

The 4194A's Auto Sequence Program (ASP) feature, an internal programming function, allows the automatic execution of measurement condition setup, compensation, measurement, calculation and display. Figure 1 shows a sample ASP program for balanced cable measurements. This program uses a unique application of the calibration function of the HP 4194A. The following is a brief discussion of programmed calibration, measurement, and analysis.

Line 170 to 290: Calibration using the 0Ω/0S/50Ω standards

Lines 300 - 360: 0S standard (HP PN 04191-85302) Calibration data input

The reference values to calculate the theoretical calibration data for each calibration standard, 0S + 0F for the 0S standard, 0Ω + 0H for the 0Ω standard, and 50Ω + 0H for the 50Ω standard, are prestored into the HP 4194A. Each time the HP 0S standard (0S + 0.082pF) is used, its data should be restored. Lines 310 to 350 perform the 0S standard frequency simulation, and line 360 inputs the simulation data into the calibration data standard.

Lines 370 to 450: Zero Open/Short offset for the HP 16093B

Performs Zero offset compensation for the fixturing from the calibration terminal (APC-7) to the test device connection terminal (HP 16093B).

Lines 460 to 860: Cable measurement

This ASP program uses the Open-Short method (based on the following equation) to calculate secondary parameters.

Characteristic Impedance:

$$|Z| = \sqrt{|Zop| \cdot |Zst|}$$

$$\theta = (\theta op + \theta st)/2$$

$|Zop|$, θop: Measured values from open measurement

$|Zst|$, θst: Measured values from short measurement

Attenuation Constant:

$$\alpha = \frac{1}{2l} \log \sqrt{\frac{(1+R)^2 + X^2}{(1-R)^2 + X^2}} \times 9865.9 \quad \text{[dB/km]}$$

Phase Constant:

$$\beta = \frac{1}{2l} \left(\pi - \arctan \frac{R+1}{X} + \arctan \frac{R-1}{X}\right) \times 1000 \quad \text{[rad/km]}$$

$$P = \sqrt{|Zst| / |Zop|}$$

$$\phi = (\theta st - \theta op)/2$$

$$R = P\cos\phi$$

$$X = P\sin\phi$$

l: Cable length [m]

Figure 1 ASP Balanced Cable Measurement Program Listing

```
10 ! ****** BALANCED CABLE MEASUREMENT ******
20 RST
30 SWT2;CMPN2;ITM2;RAD
40 START=100 KHZ
50 STOP=10 MHZ
60 R0=401
70 NOP=R0        ! NO. OF POINT
80 BEEP
90 OISP "INPUT CABLE LENGTH (m)"
100 PAUSE
110 R1=Z         ! CABLE LENGTH
120 BEEP
130 Z=0
140 OISP "NEEO COMPENSATION ? Y-->1"
150 PAUSE
160 IF Z=0 THEN GOTO 460
170 !##### CALIBRATION #####
180 BEEP
190 DISP "CONNECT 0 S"
200 PAUSE
210 CALY
220 BEEP
230 OISP "CONNECT 0 "
240 PAUSE
250 CALZ
260 BEEP
270 OISP "CONNECT STD"
280 PAUSE
290 CALSTD
300 !##### CAL OATA INPUT #####
310 IMP9
320 EQDSP
330 EQC4
340 EQVR=0;EQVL=0;EQVCA=.082E-12
350 FCHRS
360 TYG=C;TYB=D;SPA0;SPB0;CAL1
370 !##### OPEN/SHORT OFFSET #####
380 BEEP
390 DISP "OPEN"
400 PAUSE
410 ZOPEN
420 BEEP
430 DISP "SHORT"
440 PAUSE
450 ZSHRT
460 !##### CABLE MEASUREMENT #####
470 CAL1;OPN1;SHT1
480 IMP1
490 BEEP
500 DISP "CABLE OPEN MEAS !"
510 PAUSE
520 SWTRG
530 AUTOA;RE=A;RF=B
540 BEEP
550 DISP "CABLE SHORT MEAS !"
560 PAUSE
570 SWTRG
580 AUTOA;RG=A;RH=B
590 OISP ""
600 !##### CHARACTERISTIC IMPEDANCE #####
610 RA=SQR(RE*RG);RB=(RF+RH)*180/PI/2
620 CMT"CHARACTERISTIC IMPEDANCE OF CABLE"
630 OEG;A=RA;B=RB;AUTOA;BMAX=180;BMIN=-180
640 BEEP
650 DISP "PRESS CONT"
660 PAUSE
670 !##### ATT./PHASE CONSTANTS #####
680 CMT"ATTENUATION / PHASE CONSTANT"
690 DISP "CALCULATING!"
700 RAD
710 C=SQR(RG/RE);D=(RH-RF)/2
720 E=C*COS(O);F=C*SIN(O)
730 G=((1+E)*(1+E)+F*F)/((1-E)*(1-E)+F*F)
740 RC=1/2/R1*LN(SQR(G))*9685.9
750 RD=1/2/R1*(PI-ATAN((E+1)/F)+ATAN((E-1)/F))*1000
760 !    ## PHASE EXPANSION FOR PHASE CONST.
770 R22=0;RI=RD
780 FOR R3=2 TO R0
790  R2=R3-1;R11=RI(R2)-RI(R3)
800  IF R11>2*PI THEN R22=R22+R11
810  RI(R2)=RI(R3)+R22
820 NEXT R3
830 RI(R0)=RI(R0)+R22
840 BEEP
850 A=RC;B=RI;AUTO;UNIT0
860 DISP "A=DB/Km  B=RAD/Km"
870 END
```

Figure 2 Sample Measurement Results (Characteristic Impedance)



Figure 3 Sample Measurement Results (Attenuation/Phase Constant)



Figure 2 and 3 shows the results of a calculation using this ASP Program to evaluate the characteristic impedance and attenuation/phase constants of a twisted pair cable(170m long).

—— Balun Requirement ——

* The balun should have flat impedance characteristics over the required frequency range. That is, the variation in insertion loss over the frequency range should not exceed 3 dB.

* The balun's short impedance value |Zs| should be as low as possible, approximately one-tenth (or less) the characteristic impedance of the cable being tested. The lower the value of the short impedance, the smaller the additional error will be. For example, if the short impedance of the balun is one-tenth that of the cable, the additional error will be a maximum of 20% of the measurement instruments accuracy (inst. accuracy x 1.2).

* The open impedance value |Zo| of the balun, conversely, should be as high as possible, at least ten times greater than the cable's characteristic impedance.
The higher the value of the balun's open impedance, the smaller the additional error will be.

The sample balun used in Figure 4, HP PN 9100-0855, is sufficient for cable measurements at high frequencies (above 100kHz). For low frequency measurements (below 100kHz), a transformer which has a higher open impedance |Zo| at low frequencies should be used. Generally, a transformer's impedance decreases at lower frequencies and the cable's impedance increases, so errors due to |Zo| become significant. Transformers which have a higher impedance, such as those used for communications, are recommended.

Figure 4 Balun Connection



Pulse Transformer: HP PN 9100-0855 (TDK 113G1)

Standards: 0Ω: HP PN 04191-85300

0S: HP PN 04191-85302

50Ω: HP PN 04191-85301

For more information, call your local HP sales office listed in the telephone directory white pages. Ask for the Electronic Instruments Department. Or write to Hewlett-Packard: **U.S.A.**: P.O. Box 10301, Palo Alto, CA 94303-0890. **Europe**: P.O. Box 999, 1180 AZ Amsteiveen, The Netherlands. **Canada**: 6877 Goreway Drive, Mississauga, L4V 1M8, Ontario. **Japan**: Yokogawa-Hewlett-Packard Ltd., 3-29-21, Takaido-Higashi, Suginami-ku, Tokyo 168. Elsewhere in the world, write to Hewlett-Packard Intercontinental, 3495 Deer Creek Road, Palo Alto, CA 94304.

Printed in Japan November, 1986 5950-2918